
Part 2 of  Building building “stuff” in POVRay 

OK – a recap – at the end of Part 1 our POV code
“tutorial.pov” probably looks like this:
#include "colors.inc"
#include "gridsetc.inc"

// switches and stuff
#declare axes = 0.1;
#declare noshadow = 1;
#declare grid = 0.05;
#declare rd=seed(31574);
#declare var = 1;
#declare Space = < 0, 1000, 0>;

// camera location
camera {
  location  < 25.0, 22.5, -50.0>
  look_at   <0.0, 6.0,  0.0>
  angle      20
}

// illumination
light_source { < 200,400,-200> White}
light_source { <-200, 50,-400> White}

background {SkyBlue}

plane { y, 0.0 pigment{ForestGreen} }

#if (axes > 0) object{axes1(axes,noshadow) translate < 0, 0.0, 0> }
#end
#if (grid > 0) object{gridsp(grid) translate < 0, 0.0, 0> } #end  

// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ##########
// ########## PUT STUFF HERE ########## 

// ARROW SLIT CUTOUT
#declare wedgie = difference{
    box { <-1, -2, -4>, <1, 4, 0> }//pigment{Red} }
    
    box { <-1, -3, -5>, <0, 5, 1> rotate 10*y }// pigment{Blue} }
    box { < 1, -3, -5>, <0, 5, 1> rotate -10*y }// pigment{Blue} }

    box { <-2,  -3, -5>, <2, 0, 1> rotate -25*x}// pigment{Blue} }
    pigment {Grey}
}
    
    
// TOWER
difference{
    box { <-3,   0, -3> < 3, 12,  3>  }
    box { <-2,  -1, -2> < 2, 13,  2>  }
    object {wedgie scale < 0.5, 0.4, 0.5> translate < 0, 8, -1.5>}
    pigment {Grey}
}

So feel free to copy all this and paste over your own file contents here if you want.

Let's try using our wedgie for some crenellations to start
with. Firstly we need to find a good height to work at.
With a quick think I decided on 11.5 so I moved my
sample wedgie cutout there.

// TOWER
difference{
    box { <-3,   0, -3> < 3, 12,  3>  }
    box { <-2,  -1, -2> < 2, 13,  2>  }
    object {wedgie scale < 0.5, 0.4, 0.5> translate <0, 11.5, -1.5>}
    pigment {Grey}
}

We're going to need a few more of these! 
We could just copy the wedgie object line lots of times
and edit the translate coordinates. I won't deny that in
some cases I do that but in this case we're going to put
some basic software style code in there instead.

If you're an old geezer like me and spent your youth writing in old
BASIC on something like an Amiga or BBC micro, this should all be
like riding that bicycle!

The “while” statement – or “loop”. Simply put ,this is a
way of repeating something many times “while” a certain
condition remains true – and ceasing when it is untrue.
Very commonly one makes a counter and uses that to
set the number of iterations of the “loop”.

We'll use a sort of counter here that relates to our model.
We'll call the counter variable “along”
We want our wedgie to be repeated along the top of the
wall, from around -2 to +2 along the x direction.
So we initialise our counter to -2

#declare along = -2;

(You need a semi-colon at the end of most numerical declarations.)



Now our “while” statement: #while (along <= 2)

(The condition must be inside parentheses. )

We'll need to increment our counter inside the “loop” #declare along = along + 1;

(If you're not a softwarey kind of guy, this looks stupid – try to get into
the habit of seeing “=” as meaning “gets” when it is used in a
declaration! It still means “equal to” in a condition!)

Finally tell the loop where to end and go back. #end

So the full thingy is:

#declare along = -2;
#while (along <= 2)
   <DO SOMETHING CLEVER>
   #declare along = along + 1;
#end

You'll notice I've indented some of the statements – it helps with clarity
– I tend to leave the #end unindented, it's up to you.

For our case “something clever” isn't all that clever but it
works – we put the “along” variable in as the x-coordinate
of our translate command:

#declare along = -2;
#while (along <= 2)
   object {wedgie scale <0.5,0.4,0.5> translate < along,11.5,-1.5>}
   #declare along = along + 1;
#end

Like it?

It's all very well, but what about the other walls?

Well, all we want here is to copy this idea around by
rotation. Like BASIC we can nest #while loops. So
adding a rotate command  (rotating around the y-axis):

#declare ang = 0;
#while (ang < 360)
    
    #declare along = -2;
    #while (along <= 2)
       object {wedgie scale < 0.5, 0.4, 0.5> 
                      translate < along, 11.5, -1.5> 
                      rotate ang*y}
       #declare along = along + 1;
    #end  
    
#declare ang = ang + 90;
#end 

How did that work?  Well, you have to remember that each wedgie gets placed individually. We moved the wedgie
up and outwards first (and sideways) THEN rotated it, the thing rotated around the y-axis at the distance we had
set it, including any sideways offset. If we rotated the wedge first, it would “spin” close up to the y-axis and then be
moved up and outwards – all wrong!

To cut a long story short on #while loops, we can also do
something about those rather plain walls....
I only need a couple per floor and only on two walls so a
quick copy and paste and a little tinkering gives

#declare ang = 0;
#while (ang < 180)

    #declare along = -1;
    #while (along <= 1)
        object {wedgie scale < 0.5, 0.4, 0.5> 
                       translate < along, 8.0, -1.5> 
                       rotate -ang*y}
        object {wedgie scale < 0.5, 0.4, 0.5> 
                       translate < along, 5.0, -1.5> 
                       rotate -ang*y}
        #declare along = along + 2;
    #end  

#declare ang = ang + 90;
#end

We could have put in another #while loop for the two floors, but, what
the hey, copying stuff is quicker in this case I reckon.
We can't see the arrow slits on the second wall, it's on the -x side. Trust
me, they're there.



Why only two walls, well, ultimately this is just one tower for our Keep. It's gonna need some doors and stuff to get
in and out.

OK so let's make a door. Once again, we'll build it at the
origin and move it afterwards, so we'll need to hide our
tower for a bit. I could use the old /* … */ comment
markers, but since we intend that our tower be replicated
in future, why not just declare it as a CSG now?

// TOWER
#declare tower = difference{
    box { <-3,   0, -.... etc

That way, it just disappears until we put in an object
statement to call it.

New primitive : the Cylinder, very useful.
Once again we can use the Insert-Shapes thing

// Capped Cylinder, closed [or open ended]
// cylinder { <END1>, <END2>, RADIUS [open] }
//  END1 = coord of one end of cylinder
//  END2 = coord of other end
// RADIUS = size of cylinder
// open = if present, cylinder is hollow, else capped
cylinder {
  0*x,  3*x,  1
  // open
}

Hrm, this is one of those cases where comments just
drown the command!
Basically it's simpler with the shorter Insert menu option!

Cylinder { <0, 0, 0>, <3, 0, 0>, 1 }

Its just two ends and a radius. You can put “open” in
there to make it hollow, but the wall of this hollow tube
has no thickness -  unrealistic. I don't use it.

So, I'm going to use this along with another box to create
an arched door. Being generous, I'll make the door 1m
wide, 2m high and 0.1m thick.

box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> pigment{Blue} }
cylinder { <-0.05, 1.5, 0>, <0.05, 1.5, 0>, 0.5 pigment{Red}} 

OK, so it looks a bit small here but the dimensions are
“real”, as it were. I'll declare it as a CSG as it stands and
use the “ union{ “ CSG operation. I'll big it  up too,
temporarily, to illustrate some things

#declare door = union{
box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> pigment{Blue} }
cylinder { <-0.05, 1.5, 0>, <0.05, 1.5, 0>, 0.5 pigment{Red} }
}
    
object {door scale 5}



Firstly, you'll see all that speckly stuff on the surface
where the two objects share the same surfaces. POVray
has a bit of a headache here and it's best/simplest
avoided by making one or other shape a teeny weeny bit
thicker (in this case the cylinder is easiest). I'm doing this
here to illustrate a point, if the two objects were the same
colour or texture etc there isn't usually  a problem, erm. 

#declare door = union{
box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> pigment{Blue} }
cylinder {<-0.05001, 1.5, 0>,<0.05001, 1.5, 0>, 0.5 pigment{Red} }
}
    
object {door scale 5}

DIGRESSION
Anyway, what I'd like to do here is illustrate the four CSG
operations: 
If we set both cylinder and box to pigment “Bluet” - (our
transparent colour version...)

#declare door = union{
box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> pigment{Bluet} }
cylinder { <-0.050001, 1.5, 0>, <0.050001, 1.5, 0>, 0.5 pigment{Bluet}
}
}

With the “union” CSG it just jams the objects together,
much like just listing a set of primitives – but it brings all
the benefits of a CSG with it, overall colouring, texturing
declaration and so on.

If we change “union” to “merge” and render again you'll
see that “merge” works a bit harder to mash the shapes
in into one. We can see that the internal surfaces are
removed. If we wanted this to be a glass door (for some
reason????) this would be the CSG to use.

Change to “difference” and we see what we already
know.



Lastly, using “intersection”, you only see the coincident
volume of the objects.

Unless it's important, use “union” rather than “merge”,
merge generally takes longer to render.

End digression: let's have some fun with a door! Erm.

Setting all my door stuff back to “normal” again... I'm also
going to bring back the tower, make our door pigment-
less – and call up a normal sized red one moved up and
away a bit in the x direction.

#declare door = union{
    box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> }
    cylinder { <-0.050001, 1.5, 0>, <0.050001, 1.5, 0>, 0.5 }
}
 
object {door  translate < 4, 8, 0>  pigment{Red}}

object {tower}     

I'm going to use this CSG shape 3 ways and in different
colours so a “blank” uncoloured declared CSG is best
here.

Firstly, we want a surround for our doors, which will be
for access from the Keep walls (that we haven't made
yet) and at floor level...
So – make it Grey, scale it up a bit , and place it “flush”
with the outer surface of our tower (3m). Since the door
CSG has thickness symmetrical across the x=0 plane it
will stand proud a bit (depending on the x scaling factor
we give it).
(I've also tweaked one of the lights to illuminate that
slightly more)

object {door scale < 2, 1.1, 1.1>  translate < 3, 7, 0>
pigment{Grey} }

(// illumination
light_source { < 200,400,-100> White}

light_source { <-200, 50,-400> White} )

Now then, we need to think out this a bit. We want this tower to be one single #declared “thing”. Which means this
door will have to go inside the #declare tower = etc... higher up. At the moment that “thing” is a box with lots of bits
hacked out of it and now we're adding something to it again. 
What we have to think about, therefore, is how this stuff gets nested.
For now this bit is easy, we want to add this to the basic box of the tower. So we make “union” of the box and this
door surround.



We have to move our “door” declaration above the tower
one if we want to use it inside there.
By making a “union” of the basic box and our “door”
surround we make it a “single thing” at the top of the list
of things in the difference statement. 

#declare door = union{
    box { <-0.05, 0, -0.5>, <0.05, 1.5, 0.5> }
    cylinder { <-0.050001, 1.5, 0>, <0.050001, 1.5, 0>, 0.5 }
}
    
// TOWER
#declare tower = difference{
    
    // "stuff we add to the basic box of the tower"
    union{
        box { <-3,   0, -3> < 3, 12,  3>  }
        object {door scale < 2, 1.1, 1.1>  translate < 3, 7, 0>  }
    }

    // "stuff we "chop off" the basic tower
    box { <-2,  -1, -2> < 2, 13,  2>  }
   

    #declare ang ….etc

Notice any difference? (I hope not!)

OK, so it's no kind of door without a hole in it. So now
that the door surround is part of the union we can chop
lumps out of, we just add a door shape (stretched in the x
direction) to our list of difference items...
// TOWER
#declare tower = difference{
    
    // "stuff we add to the basic box of the tower"
    union{
        box { <-3,   0, -3> < 3, 12,  3>  }
        object {door scale < 2, 1.1, 1.1>  translate < 3, 7, 0> }
    }

    // "stuff we "chop off" the basic tower
    box { <-2,  -1, -2> < 2, 13,  2>  }

.

.
        ...lare ang = ang + 90;
    #end 
    
    object {door scale < 25252525, 1, 1>  translate < 3, 7, 0> }
    
    pigment {Grey}
}

Handy Tip:  If you're modifying just a small part of your
scene and you don't want/need to render the whole view
you can render just the area you're interested in. If you
have an existing render window open, left click and hold
then sweep a box area of the render screen you want to
see then release. A dialogue will pop up ask you to
confirm, do so.
Render time is much reduced.

Now then. We want to put a door inside the door cut-out.
We can't add this to our basic tower/door union, it would
get “chopped” out by the difference. This is something we
need to add “after” the difference occurs.
What we actually need to do is demote our difference
block and make the tower #declare a union of this new
door and the tower with its cutouts.
[Think of something like a beaker – it's a cylinder with
another cylinder removed from inside. If we then put
liquid inside it has to happen outside of the action of
removing the inner cylinder.]

So we have another level of indenting and our main
declaration is a union. More often than not, this is what
most of my things look like. 
I'll also say that there are probably other ways of doing
this, this is just the way I do them!

// TOWER
#declare tower = union {

    difference{
        
        // "stuff” we add to the basic box of the tower
        union{
            box { <-3,   0, -3> < 3, 12,  3>  }
            object {door scale < 2, 1.1, 1.1>  translate < 3, 7, 0> }
        }  // end of basic tower "adds"
    
        // "stuff we "chop off" the basic tower
        box { <-2,  -1, -2> < 2, 13,  2>  }
.
(clipped)
.
        object {door scale < 25, 1, 1>  translate < 3, 7, 0> }
        
        pigment {Grey}
    } // end of chop off stuff
 
} //end of TOWER declare object



Since out main #declare is a “union” there is no particular
“order” in which our  two things need to go. The two
objects – the difference containing most of the tower
building , and our new door, can be in either order. By
convention I usually put the “outside adds” at the end...
So
.
.
        ..ment {Grey}
    } // end of chop off stuff
    
    //actual door 
    object {door scale < 0.05, 1, 1>  translate < 2.8, 7, 0> pigment
{Brown} }

}  //end of TOWER declare object

We want 3 of these in all, two at the wall top level and
one at ground level so we need to replicate the “adds” for
the surrounds, the cut-outs and the wood door.

// "stuff we add to the basic box of the tower"
union{
    box { <-3,   0, -3> < 3, 12,  3>  }
    object {door scale < 2, 1.1, 1.1> translate < 3, 7, 0> }
    object {door scale < 2, 1.1, 1.1> translate < 3, 7, 0> rotate
-90*y}
    object {door scale < 2, 1.1, 1.1> translate < 3, 0, 0> }
} // end of basic tower "adds"

and later
        #declare ang = ang + 90;
        #end 
        
        object {door scale < 25, 1, 1>  translate < 3, 7, 0> }
        object {door scale < 25, 1, 1>  translate < 3, 7, 0> rotate
-90*y }
        object {door scale < 25, 1, 1>  translate < 3, 0, 0> }
        
        pigment {Grey}
    } // end of chop off stuff
    
    //actual door 
    object {door scale < 0.05, 1, 1>  translate < 2.9, 7, 0> pigment
{Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 7, 0>  rotate
-90*y pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 0, 0> pigment
{Brown} }

}  //end of TOWER declare object

Now, you could just take my word for it that there is a
third door round back, at wall height, or you could change
the viewpoint – OR in this situation, just rotating our
tower 180 degrees is simpler?

object {tower rotate 180*y}

As you can see, since the Tower is a single “declared”
thing we can do stuff to it as a single “thing”.

OK – so I just spotted something. Let's zoom in on that
door by changing our camera's “angle” parameter:

// camera location
camera {
  location  < 25.0, 22.5, -50.0>
  look_at   <0.0, 6.0,  0.0>
  angle      10
}

That surround is not really all that equal all around. I'd
like to change that.



So I did some “tweaking” - not uncommon for me!

I centred the door - rather than having go from 0 up to 2m, I set it for -1 to +1.

Then I moved all instances of the door, surrounds and cut-outs up 1m. I also widened the surround scaling factor a
bit 

// arched door, 1m wide, 2m high. Translate +1*y to put base at floor level
#declare door = union{
    box { <-0.05, -1, -0.5>, <0.05, 0.5, 0.5> }
    cylinder { <-0.050001, 0.5, 0>, <0.050001, 0.5, 0>, 0.5 }
}
 
    
// TOWER
#declare tower = union {

    difference{
        
// "stuff we add to the basic box of the tower"
union{
    box { <-3,   0, -3> < 3, 12,  3>  }
    object {door scale < 2, 1.1, 1.2>  translate < 3, 8, 0> }
    object {door scale < 2, 1.1, 1.2>  translate < 3, 8, 0> rotate -90*y}
    object {door scale < 2, 1.1, 1.2>  translate < 3, 1, 0> }
} // end of basic tower "adds"
    
        // "stuff we "chop off" the basic tower
        box { <-2,  -1, -2> < 2, 13,  2>  }
       
        #declare ang = 0;
        #while (ang < 360)
                #declare along = -2;
                #while (along <= 2)
                    object {wedgie scale < 0.5, 0.4, 0.5> 
                                   translate < along, 11.5, -1.5> 
                                   rotate ang*y}
                    #declare along = along + 1;
                #end  
        #declare ang = ang + 90;
        #end
    
    
        #declare ang = 0;
        #while (ang < 180)
            #declare along = -1;
            #while (along <= 1)
                object {wedgie scale < 0.5, 0.4, 0.5> 
                               translate < along, 8.0, -1.5> 
                               rotate ang*y}
                object {wedgie scale < 0.5, 0.4, 0.5> 
                               translate < along, 5.0, -1.5> 
                               rotate ang*y}
                #declare along = along + 2;
            #end  
        #declare ang = ang + 90;
        #end 
        
        object {door scale < 25, 1, 1>  translate < 3, 8, 0> }
        object {door scale < 25, 1, 1>  translate < 3, 8, 0> rotate -90*y }
        object {door scale < 25, 1, 1>  translate < 3, 1, 0> }
        
        pigment {Grey}
    } // end of chop off stuff
    
    //actual door 
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0> pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0>  rotate -90*y pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 1, 0> pigment {Brown} }

}  //end of TOWER declare object

And got this, which I think looks better. The bottom of the
door is still at a unit floor level, so if we put in our wall
walkway at 7m, the bottom part of the surround will be
“absorbed” / hidden.



So, we'll go back to our original view by going back to
camera angle 20, and removing the rotate from the tower
call. However, I'd like to look down on the top of the
tower now, so I set our camera thus

// camera location
camera {
  location  < 25.0, 50, -50.0>
  look_at   <0.0, 6.0,  0.0>
  angle      20
}

Why? Well, the poor guys on the fighting tops have
nothing to stand on. So, along with our wood doors, we
need to add a floor about 1m below the bottom of our
crenellations – outside the difference part of the tower. A
simple box is all I'm going to need here

    } // end of chop off stuff
    
    //actual door 
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0> pigment
{Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 8, 0>  rotate
-90*y pigment {Brown} }
    object {door scale < 0.05, 1, 1>  translate < 2.9, 1, 0> pigment
{Brown} }
    
    // top floor
    box { <-2.5, -0.1, -2.5>, <2.5, 0, 2.5> translate 10.5*y pigment
{Brown}}

}  //end of TOWER declare object.

It's 2.5m square so it intrudes into our walls and 10cm
thick. I've made it go from -0.1y to 0y so that floor
“height” isn't at some weird and wonderful decimal point
value... ehem well... that said, it's at 10.5m, but the
number is nice and obvious in the translate command,
not buried in the box parameters themselves.

We could add a hatch and all sorts of stuff, but I'd be
repeating the lesson here somewhat. So we'll keep it
simple.

OK. That's probably about it for now. I think I've
introduced some POV concepts here and that's the point
here, not creating a masterpiece!

In the meantime, you now, hopefully, have all the
knowledge you need to create a spiral (ehem, helical)
staircase. Have a think about it!


